

# kan

including double coset rewriting systems

1.28

4 February 2018

**Anne Heyworth**

**Christopher D. Wensley**

**Christopher D. Wensley**

Email: [c.d.wensley@bangor.ac.uk](mailto:c.d.wensley@bangor.ac.uk)

Homepage: <http://pages.bangor.ac.uk/~mas023/>

Address: Dr. C.D. Wensley

School of Computer Science

Bangor University

Dean Street

Bangor

Gwynedd LL57 1UT

UK

## Abstract

Kan is a GAP package originally implemented in 1996 using the GAP 3 language, to compute induced actions of categories, when the first author was studying for a Ph.D. in Bangor.

This reduced version only provides functions for the computation of normal forms of representatives of double cosets of finitely presented groups.

Kan became an accepted GAP package in May 2015.

Bug reports, suggestions and comments are, of course, welcome. Please contact the last author at [c.d.wensley@bangor.ac.uk](mailto:c.d.wensley@bangor.ac.uk) or submit an issue at the GitHub repository <https://github.com/gap-packages/kan/issues/>.

## Copyright

© 1996-2018 Anne Heyworth and Chris Wensley

The Kan package is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version.

## Acknowledgements

This documentation was prepared with the GAPDoc [LN17] and AutoDoc [GH17] packages.

The procedure used to produce new releases uses the package GitHubPagesForGAP [Hor17] and the package ReleaseTools.

# Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
<b>2</b>	<b>Double Coset Rewriting Systems</b>	<b>6</b>
2.1	Rewriting Systems . . . . .	6
2.2	Example 2 – free product of two cyclic groups . . . . .	7
2.3	Example 3 – the trefoil group . . . . .	10
2.4	Example 4 – an infinite rewriting system . . . . .	13
<b>3</b>	<b>Development History</b>	<b>17</b>
3.1	Versions of the package . . . . .	17
3.2	What needs doing next? . . . . .	17
	<b>References</b>	<b>19</b>
	<b>Index</b>	<b>20</b>

# Chapter 1

## Introduction

The Kan package started out as part of Anne Heyworth's thesis [Hey99], and was designed to compute induced actions of categories (see also [BH00]).

This version of Kan only provides functions for the computation of normal forms of representatives of double cosets of finitely presented groups, and is made available in support of the paper [BGHW06]. Existing methods for computing double cosets in GAP are described in [Lin91].

The package is loaded into GAP with the command

Example

```
gap> LoadPackage( "kan" );
```

The package may be obtained as a compressed tar file `kan-version.number.tar.gz` by ftp from one of the following sites:

- the Kan GitHub site: <https://github.com/gap-packages.github.io/kan/>.
- any GAP archive, e.g. <https://www.gap-system.org/Packages/packages.html>;

The package also has a GitHub repository at: <https://github.com/gap-packages/kan/>.

Some of the functions in the `automata` package are used to compute word acceptors and regular expressions for the languages they accept.

The `KBMag` package is also used to compute a word acceptor of a group  $G$  when  $G$  has no finite rewriting system. If `KBMag` is not available (the user is not on a UNIX system, or the C-programs have not been compiled), the file `dckbmag.gi` will not be read, so methods for the functions detailed in section 2.4.1 will not be available.

Once the package is loaded, it is possible to check the installation is correct by running a test file of the manual examples with the following command. (The test file itself is `tst/fulltest.tst` or `tst/parttest.tst`, depending whether or not `KBMag` is available.)

Example

```
gap> ReadPackage( "kan", "tst/testall.g" );
#I Testing /Applications/gap/my-dev/pkg/kan/tst/fulltest.tst
#I No errors detected while testing package kan
true
```

The information parameter `InfoKan` takes default value 0. When raised to a higher value, additional information is printed out.

Once the package is loaded, the manual `doc/manual.pdf` can be found in the documentation folder. The html versions, with or without MathJax, may be rebuilt as follows.

Example

```
gap> InfoLevel( InfoKan );
0
gap> ReadPackage( "kan, "makedoc.g" );
```

Please send bug reports, suggestions and other comments to the second author, or use the GitHub issue tracker at <https://github.com/gap-packages/kan/issues/new>.

Additional information can be found on the *Computational Higher-dimensional Discrete Algebra* website at <http://pages.bangor.ac.uk/~mas023/chda/>.

## Chapter 2

# Double Coset Rewriting Systems

The `Kan` package provides functions for the computation of normal forms for double coset representatives of finitely presented groups. The first version of the package was released to support the paper [BGHW06], which describes the algorithms used in this package.

## 2.1 Rewriting Systems

### 2.1.1 KnuthBendixRewritingSystem

- ▷ `KnuthBendixRewritingSystem(grp, gensorder, ordering, alph)` (operation)
- ▷ `ReducedConfluentRewritingSystem(grp, gensorder, ordering, lim, alph)` (operation)
- ▷ `DisplayRwsRules(rws)` (operation)

Methods for `KnuthBendixRewritingSystem` and `ReducedConfluentRewritingSystem` are supplied which apply to a finitely presented group. These start by calling `IsomorphismFpMonoid` and then work with the resulting monoid. The parameter `ordering` will normally be "shortlex" or "wreath", while `gensorder` is an integer list for reordering the generators, and `alph` is an alphabet string used when printing words. A *partial* rewriting system may be obtained by giving a limit to the number of rules calculated. As usual,  $A, B$  denote the inverses of  $a, b$ .

We take as an example the fundamental group of the oriented surface of genus 2. The generators are by default ordered  $[A, a, B, b, C, c, D, d]$ , so the list  $L = [2, 1, 4, 3, 6, 5, 8, 7]$  is used to specify the order  $[a, A, b, B, c, C, d, D]$  to be used with the wreath ordering. Specifying a limit 0 means that no limit is prescribed.

The operation `DisplayRwsRules` prints out the rules using the letters in the given alphabet `alph4` rather than using the generators of the monoid.

An additional method for `ReducedForm(G, g)` is provided for a finitely presented group  $G$  with a rewriting system and an element  $g$  of  $G$ .

Example

```
gap> F4 := FreeGroup( 4 );;
gap> rels := [ Comm(F4.1,F4.2) * Comm(F4.3,F4.4) ];;
gap> H4 := F4/rels;;
gap> L := [2,1,4,3,6,5,8,7];;
gap> order := "wreath";;
gap> alph4 := "aAbBcCdD";;
```

```

gap> rws4 := ReducedConfluentRewritingSystem( H4, L, order, 0, alph4 );;
gap> DisplayRwsRules( rws4 );
[ [ aA, id ], [ Aa, id ], [ bB, id ], [ Bb, id ], [ cC, id ], [ Cc, id ], [ dD\
, id ], [ Dd, id ], [ cd, dcBAba ], [ cBAbaD, Dc ], [ CD, BAbaDC ], [ Cd, dABa\
bC ] ]
true
gap> a := H4.1;; b := H4.2;; c := H4.3;; d := H4.4;;
gap> ReducedForm( H4, c*d);
f4*f3*f2^-1*f1^-1*f2*f1

```

## 2.1.2 NextWord

- ▷ NextWord(*rws*, *w*, *limit*) (operation)
- ▷ NextWords(*rws*, *w*, *length*, *limit*) (operation)

The NextWord operation finds the next recognizable word after *w* using the rewriting system *rws*. The third parameter is the maximum number of words that will be tested before giving up. (If no limit is provided the number 100,000 is used.)

The NextWords operation applies NextWord repeatedly and returns a list of the specified length of recognizable words. (If, at any stage, the limit is reached, a truncated list is returned.)

Example

```

gap> free4 := FreeMonoidOfRewritingSystem( rws4 );;
gap> gens4 := GeneratorsOfMonoid( free4 );
[ f1, f1^-1, f2, f2^-1, f3, f3^-1, f4, f4^-1 ]
gap> NextWord( rws4, gens4[5]*gens4[7] );
f3*f4^-1
gap> NextWords( rws4, last, 20, 100 );
[ f3^-1*f1, f3^-1*f1^-1, f3^-1*f2, f3^-1*f2^-1, f3^-1^2, f4*f1, f4*f1^-1,
  f4*f2, f4*f2^-1, f4*f3, f4*f3^-1, f4^2, f4^-1*f1, f4^-1*f1^-1, f4^-1*f2,
  f4^-1*f2^-1, f4^-1*f3, f4^-1*f3^-1, f4^-1^2, f1^3 ]

```

## 2.2 Example 2 – free product of two cyclic groups

### 2.2.1 DoubleCosetRewritingSystem

- ▷ DoubleCosetRewritingSystem(*grp*, *genH*, *genK*, *rws*) (function)
- ▷ IsDoubleCosetRewritingSystem(*dcrws*) (property)

A *double coset rewriting system* for the double cosets  $H \setminus G / K$  requires as data a finitely presented group  $G = \text{grp}$ ; generators *genH*, *genK* for subgroups  $H, K$ ; and a rewriting system *rws* for  $G$ .

A simple example is given by taking  $G$  to be the free group on two generators  $a, b$ , a cyclic subgroup  $H$  with generator  $a^6$ , and a second cyclic subgroup  $K$  with generator  $a^4$ . (Similar examples using different powers of  $a$  are easily constructed.) Since  $\gcd(6, 4) = 2$ , we have  $Ha^2K = HK$ , so the double cosets have representatives  $[HK, HaK, Ha^i b a^j K, Ha^i b w b a^j K]$  where  $i \in [0..5]$ ,  $j \in [0..3]$ , and  $w$  is any word in  $a, b$ .

In the example the free group  $G$  is converted to a four generator monoid with relations defining the inverse of each generator,  $[[Aa, id], [aA, id], [Bb, id], [bB, id]]$ , where  $id$  is the empty word. The initial rules for the double coset rewriting system comprise those of  $G$  plus those given by the generators of  $H, K$ , which are  $[[Ha^6, H], [a^4K, K]]$ . In the complete rewrite system new rules involving  $H$  or  $K$  may arise, and there may also be rules involving both  $H$  and  $K$ .

For this example,

- there are two  $H$ -rules,  $[[Ha^4, HA^2], [HA^3, Ha^3]]$ ,
- there are two  $K$ -rules,  $[[a^3K, AK], [A^2K, a^2K]]$ ,
- and there are two  $H$ - $K$ -rules,  $[[Ha^2K, HK], [HAK, HaK]]$ .

Here is how the computation may be performed.

Example

```
gap> G1 := FreeGroup( 2 );;
gap> L1 := [2,1,4,3];;
gap> order := "shortlex";;
gap> alph1 := "AaBb";;
gap> rws1 := ReducedConfluentRewritingSystem( G1, L1, order, 0, alph1 );
Rewriting System for Monoid( [ f1, f1^-1, f2, f2^-1 ], ... ) with rules
[ [ f1*f1^-1, <identity ...> ], [ f1^-1*f1, <identity ...> ],
  [ f2*f2^-1, <identity ...> ], [ f2^-1*f2, <identity ...> ] ]
gap> DisplayRwsRules( rws1 );;
[ [ Aa, id ], [ aA, id ], [ Bb, id ], [ bB, id ] ]
gap> genG1 := GeneratorsOfGroup( G1 );;
gap> H1 := Subgroup( G1, [ genG1[1]^6 ] );;
gap> K1 := Subgroup( G1, [ genG1[1]^4 ] );;
gap> dcrws1 := DoubleCosetRewritingSystem( G1, H1, K1, rws1 );;
gap> IsDoubleCosetRewritingSystem( dcrws1 );
true
gap> DisplayRwsRules( dcrws1 );;
G-rules:
[ [ Aa, id ], [ aA, id ], [ Bb, id ], [ bB, id ] ]
H-rules:
[ [ Haaaa, HAA ],
  [ HAAA, Haaa ] ]
K-rules:
[ [ aaaK, AK ],
  [ AAK, aaK ] ]
H-K-rules:
[ [ HaaK, HK ],
  [ HAK, HaK ] ]
```

## 2.2.2 DisplayAsString

▷ `DisplayAsString(word, alph)`

(operation)



This operation displays a double coset using letters of the alphabet obtained by concatenating "HK" with the alphabet for the monoid obtained above. In the example a double coset  $w$  and its reduced form  $rw$  are displayed.

Example

```
gap> free := FreeMonoidOfRewritingSystem( dcrws1 );;
gap> mon := MonoidOfRewritingSystem( dcrws1 );;
gap> gens := GeneratorsOfMonoid( free );;
gap> H := gens[1];; K := gens[2];;
gap> A := gens[3];; a := gens[4];;
gap> B := gens[5];; b := gens[6];;
gap> alph2 := Concatenation( "HK", alph1 );
"HKAAaBb"
gap> w := H*a^5*b^3*a^5*K;
m1*m4^5*m6^3*m4^5*m2
gap> DisplayAsString( w, alph2 );
HaaaaabbbaaaaaK
gap> rw := ReducedForm( dcrws1, w );
m1*m3*m6^3*m4*m2
gap> DisplayAsString( rw, alph2 );
HAbbbaK
```

### 2.2.3 WordAcceptorOfReducedRws

- ▷ WordAcceptorOfReducedRws(*rws*) (attribute)
- ▷ WordAcceptorOfDoubleCosetRws(*rws*) (attribute)
- ▷ IsWordAcceptorOfDoubleCosetRws(*aut*) (property)

Using functions from the `automata` package, we may

- compute a *word acceptor* for the rewriting system of  $G$ ;
- compute a *word acceptor* for the double coset rewriting system;
- test a list of words to see whether they are recognised by the automaton;
- obtain a rational expression for the language of the automaton.

Example

```
gap> waG1 := WordAcceptorOfReducedRws( rws1 );
Automaton("det",6,"aAbB",[ [ 1, 4, 1, 4, 4, 4 ], [ 1, 3, 3, 1, 3, 3 ], [ 1, 2,\
  2, 2, 1, 2 ], [ 1, 1, 5, 5, 5, 5 ] ],[ 6 ],[ 2, 3, 4, 5, 6 ]);;
gap> wadcl := WordAcceptorOfDoubleCosetRws( dcrws1 );
< deterministic automaton on 6 letters with 15 states >
gap> Print( wadcl );
Automaton("det",15,"HKaAbB",[ [ 2, 2, 2, 6, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2 ],\
  [ 2, 2, 1, 2, 1, 1, 2, 1, 1, 2, 2, 1, 1, 2, 2 ], [ 2, 2, 13, 2, 10, 5, 2, 13,\
  2, 7, 11, 11, 12, 2, 2 ], [ 2, 2, 9, 2, 2, 14, 2, 9, 15, 2, 2, 2, 2, 7, 15 ],\
  [ 2, 2, 2, 2, 8, 8, 8, 8, 8, 8, 8, 8, 8, 8 ], [ 2, 2, 3, 2, 3, 3, 3, 2, 3,\
  3, 3, 3, 3, 3, 3 ] ],[ 4 ],[ 1 ]);;
```

```

gap> words1 := [ "HK", "HaK", "HbK", "HAK", "HaaK", "HbbK", "HabK", "HbaK", "HbaabK" ];
gap> valid1 := List( words1, w -> IsRecognizedByAutomaton( wadcl, w ) );
[ true, true, true, false, false, true, true, true, true ]
gap> lang1 := FAtoRatExp( wadcl );
((H(aaaUAA)BUH(a(aBUB)UABUB))(a(a(aa*BUB)UB)UA(AA*BUB)UB)*(a(a(aa*bUb)Ub)UA(AA\
*bUb))UH(aaaUAA)bUH(a(abUb)UAbUb))((a(a(aa*BUB)UB)UA(AA*BUB))(a(a(aa*BUB)UB)UA\
(AA*BUB)UB)*(a(a(aa*bUb)Ub)UA(AA*bUb))Ua(a(aa*bUb)Ub)UA(AA*bUb)Ub)*((a(a(aa*BU\
B)UB)UA(AA*BUB))(a(a(aa*BUB)UB)UA(AA*BUB)UB)*(a(aKUK)UAKUK)Ua(aKUK)UAKUK)U(H(a\
aaUAA)BUH(a(aBUB)UABUB))(a(a(aa*BUB)UB)UA(AA*BUB)UB)*(a(aKUK)UAKUK)UH(aKUK)

```

## 2.3 Example 3 – the trefoil group

### 2.3.1 PartialDoubleCosetRewritingSystem

- ▷ PartialDoubleCosetRewritingSystem(*grp*, *Hgens*, *Kgens*, *rws*, *limit*) (operation)
- ▷ WordAcceptorOfPartialDoubleCosetRws(*grp*, *prws*) (attribute)

It may happen that, even when  $G$  has a finite rewriting system, the double coset rewriting system is infinite. This is the case with the trefoil group  $T$  with generators  $[c, d]$  and relator  $[c^3 = v^2]$  when the wreath product ordering is used with  $C > c > D > d$ . The group itself has a rewriting system with just 6 rules.

Example

```

gap> FT := FreeGroup( 2 );;
gap> relsT := [ FT.1^3*FT.2^-2 ];;
gap> T := FT/relsT;;
gap> genT := GeneratorsOfGroup( T );;
gap> U := Subgroup( T, [ genT[1] ] );;
gap> V := Subgroup( T, [ genT[2] ] );;
gap> alphT := "cCdD";;
gap> ordT := [3,4,1,2];;
gap> orderT := "wreath";;
gap> rwsT := ReducedConfluentRewritingSystem( T, ordT, orderT, 0, alphT );;
gap> DisplayRwsRules( rwsT );;
[ [ dD, id ], [ Dd, id ], [ C, ccDD ], [ ccc, dd ], [ ddc, cdd ], [ Dc, dcDD ] \
]
gap> accT := WordAcceptorOfReducedRws( rwsT );
< deterministic automaton on 4 letters with 7 states >
gap> Print( "accT = ", accT );
accT = Automaton("det",7,"dDcC",[ [ 6, 2, 2, 4, 6, 4, 6 ], [ 3, 2, 3, 2, 3, 2, \
3 ], [ 7, 2, 2, 2, 2, 7, 5 ], [ 2, 2, 2, 2, 2, 2, 2 ] ],[ 1 ],[ 1, 3, 4, 5, 6 \
, 7 ]);;
gap> langT := FAtoRatExp( accT );
(dcUc)((cdUd)c)*((cdUd)(dd*U@)Uc(DD*U@)UDD*U@)Ud(dd*U@)UDD*U@

```

In the following example we reduce the word  $w = d^5c^5$  to  $dc^2d^6$  and check that only the latter is recognised by the automaton `accT`.

## Example

```

gap> free := FreeMonoidOfRewritingSystem( rwsT );
gap> gens := GeneratorsOfMonoid( free );
gap> c := gens[1];; C := gens[2];; d := gens[3];; D := gens[4];;
gap> w := d^5*c^5;
f2^5*f1^5
gap> sw := WordToString( w, alphT );
"ddddccccc"
gap> IsRecognizedByAutomaton( accT, sw );
false
gap> rw := ReducedForm( rwsT, w );
f2*f1^2*f2^6
gap> srw := WordToString( rw, alphT );
"dccdddd"
gap> IsRecognizedByAutomaton( accT, srw );
true

```

In earlier versions of Kan, from about 1.01 up to 1.21, the complementary automaton and language were returned in the example above. This error has now been rectified.

In even earlier versions of Kan (in 0.95 for example) a shorter rational expression for the language was obtained from Automata. In what follows, we check that the two expressions define the same language.

## Example

```

gap> alph := AlphabetOfRatExpAsList( langT );;
gap> a1 := RatExp0nnLetters( alph, [ ], [1] );; ## y
gap> a2 := RatExp0nnLetters( alph, [ ], [2] );; ## Y
gap> a3 := RatExp0nnLetters( alph, [ ], [3] );; ## x
gap> a4 := RatExp0nnLetters( alph, [ ], [4] );; ## X
gap> s1 := RatExp0nnLetters( alph, "star", a1 );; ## y*
gap> s2 := RatExp0nnLetters( alph, "star", a2 );; ## Y*
gap> a1a3 := RatExp0nnLetters( alph, "product", [ a1, a3 ] );; ## yx
gap> u1 := RatExp0nnLetters( alph, "union", [ a1a3, a3 ] );; ## yxUx
gap> a3a1 := RatExp0nnLetters( alph, "product", [ a3, a1 ] );; ## xy
gap> u2 := RatExp0nnLetters( alph, "union", [ a3a1, a1 ] );; ## xyUy
gap> u2a3 := RatExp0nnLetters( alph, "product", [ u2, a3 ] );; ## (xyUy)x
gap> su2a3 := RatExp0nnLetters( alph, "star", u2a3 );; ## ((xyUy)x)*
gap> u2s1 := RatExp0nnLetters( alph, "product", [ u2, s1 ] );; ## (xyUy)y*
gap> a3s2 := RatExp0nnLetters( alph, "product", [ a3, s2 ] );; ## xY*
gap> u3 := RatExp0nnLetters( alph, "union", [u2s1,a3s2,s2] );;
gap> prod := RatExp0nnLetters( alph, "product", [u1,su2a3,u3] );;
gap> a1s1 := RatExp0nnLetters( alph, "product", [ a1, s1 ] );; ## yy*
gap> r := RatExp0nnLetters( alph, "union", [ prod, a1s1, s2 ] );;
(yxUx)((xyUy)x)*((xyUy)y*UxY*UY*)Uyy*UY*
gap> AreEqualLang( langT, r );
true

```

If we now take subgroups  $H = \langle c \rangle$  and  $K = \langle d \rangle$  we find that the double coset rewriting system has an infinite number of  $H$ -rules. It turns out that only a finite number of these are needed to produce

the required automaton. The function `PartialDoubleCosetRewritingSystem` allows a limit to be specified on the number of rules to be computed. In the listing below a limit of 20 is used, but in fact 10 is sufficient.

Example

```
gap> prwsT := PartialDoubleCosetRewritingSystem( T, U, V, rwsT, 20 );;
#I WARNING: reached supplied limit 20 on number of rules
gap> DisplayRwsRules( prwsT );
G-rules:
[ [ C, ccDD ], [ dD, id ], [ Dc, dcDD ], [ Dd, id ], [ ccc, dd ], [ ddc, cdd ] \
]
H-rules:
[ [ Hc, H ],
  [ HD, Hd ],
  [ Hdd, H ],
  [ Hdcdd, Hdc ],
  [ HdcD, HdcD ],
  [ Hdcdcdd, Hdcdc ],
  [ Hdccdd, Hdcc ],
  [ HdccD, HdccD ],
  [ HdcdcD, HdcdcD ],
  [ Hdcdcdcdd, Hdcdcdc ],
  [ Hdcdccdd, Hdcdcc ],
  [ Hdccdcdd, Hdccdc ],
  [ HdccdcDD, Hdccdc ] ]
K-rules:
[ [ dK, K ],
  [ DK, K ] ]
```

This list of partial rules is then used to produce a modified word acceptor function.

We then construct the double coset  $Hd^5c^5K$  and find its reduced form (compare this with the earlier example).

Example

```
gap> paccT := WordAcceptorOfPartialDoubleCosetRws( T, prwsT );;
< deterministic automaton on 6 letters with 6 states >
gap> Print( paccT, "\n" );
Automaton("det",6,"HKyYxX",[ [ 2, 2, 2, 6, 2, 2 ], [ 2, 2, 1, 2, 2, 1 ], [ 2, \
2, 5, 2, 2, 5 ], [ 2, 2, 2, 2, 2, 2 ], [ 2, 2, 6, 2, 3, 2 ], [ 2, 2, 2, 2, 2, \
2 ] ],[ 4 ],[ 1 ]);;
gap> plangT := FAtoRatExp( paccT );
H(yx(yx)*x)*(yx(yx)*KUK)
gap> wordsT := ["HK", "HxK", "HyK", "HYK", "HyxK", "HyxxK", "HyyH", "HyxYK"];;
gap> validT := List( wordsT, w -> IsRecognizedByAutomaton( paccT, w ) );
[ true, false, false, false, true, true, false, false ]

gap> pfree := FreeMonoidOfRewritingSystem( prwsT );;
gap> pgens := GeneratorsOfMonoid( pfree );;
gap> H := pgens[1];; K := pgens[2];;
gap> c := pgens[3];; C := pgens[4];;
gap> d := pgens[5];; D := pgens[6];;
```

```

gap> palphT := Concatenation( "HK", alphT );
"HKcCdD"
gap> pw := H*d^5*c^5*K;; DisplayAsString( pw, palphT );
Hddddccccck
gap> rpw := ReducedForm( prwsT, pw );;
gap> spw := WordToString( rpw, palphT );
"HdccK"
gap> ok := IsRecognizedByAutomaton( paccT, spw );
true

```

## 2.4 Example 4 – an infinite rewriting system

### 2.4.1 KBMagRewritingSystem

- ▷ KBMagRewritingSystem(*fpgrp*) (attribute)
- ▷ KBMagWordAcceptor(*fpgrp*) (attribute)
- ▷ KBMagFSAtoAutomataDFA(*fsa*, *alph*) (operation)
- ▷ WordAcceptorByKBMag(*grp*, *alph*) (operation)
- ▷ WordAcceptorByKBMagOfDoubleCosetRws(*grp*, *dcrws*) (operation)

When the group  $G$  has an infinite rewriting system, the double coset rewriting system will also be infinite. In this case we may use the function `KBMagWordAcceptor` which calls `KBMag` to compute a word acceptor for  $G$ , and `KBMagFSAtoAutomataDFA` to convert this to a deterministic automaton as used by the `automata` package. The resulting `dfa` forms part of the double coset automaton, together with sufficient  $H$ -rules,  $K$ -rules and  $H$ - $K$ -rules found by the function `PartialDoubleCosetRewritingSystem`. (Note that these five attributes and operations will not be available if the `KBMag` package has not been loaded.)

In the following example we take a two generator group  $G_4$  with relators  $[a^3, b^3, (a * b)^3]$ , the normal forms of whose elements are some of the strings with  $a$  or  $a^{-1}$  alternating with  $b$  or  $b^{-1}$ . The automatic structure computed by `KBMag` has a word acceptor with 17 states.

Example

```

gap> F4 := FreeGroup("a","b");;
gap> rels4 := [ F4.1^3, F4.2^3, (F4.1*F4.2)^3 ];;
gap> G4 := F4/rels4;;
gap> alph4 := "AaBb";;
gap> waG4 := WordAcceptorByKBMag( G4, alph4 );;
gap> Print( waG4, "\n");
Automaton("det",18,"aAbB",[ [ 2, 18, 18, 8, 10, 12, 13, 18, 18, 18, 18, 18, 18\
, 8, 17, 12, 18, 18 ], [ 3, 18, 18, 9, 11, 9, 12, 18, 18, 18, 18, 18, 11, \
18, 11, 18, 18 ], [ 4, 6, 6, 18, 18, 18, 18, 18, 6, 12, 16, 18, 12, 18, 18, 18\
, 12, 18 ], [ 5, 5, 7, 18, 18, 18, 18, 14, 15, 5, 18, 18, 7, 18, 18, 18, 15, 1\
8 ] ],[ 1 ],[ 1 .. 17 ]]);;
gap> langG4 := F4toRatExp( waG4 );
((abUAb)AUbA)(bA)*(b(aU@)UB(aB)*(a(bU@)U@)U@)U(abUAb)(aU@)U((aBUB)(aB)*AUba(Ba\
)*BA)(bA)*(b(aU@)U@)U(aBUB)(aB)*(a(bU@)U@)Uba(Ba)*(BU@)UbUaUA(B(aB)*(a(bU@)UAU\
@)U@)U@

```



```

[ HaB, Hb ]
[ Hab, H ]
[ HbAB, HAba ]
[ HbAbAB, HAbAba ]
[ HbAbAbAB, HAbAbAba ]
[ HbAbAbAbAB, HAbAbAbAba ]
[ HbAbAbAbAbAB, HAbAbAbAbAba ]
[ HbAbAbAbAbAbAB, HAbAbAbAbAbAba ]
gap> Print("K-rules:\n"); DisplayAsString( Krules(dcrws4), alph4e, true );
K-rules:
[ BK, aK ]
[ BaK, bK ]
[ baK, K ]
[ BAbK, abAK ]
[ BAbAbK, abAbAK ]
[ BAbAbAbK, abAbAbAK ]
[ BAbAbAbAbK, abAbAbAbAK ]
[ BAbAbAbAbAbK, abAbAbAbAbAK ]
[ BAbAbAbAbAbAbK, abAbAbAbAbAbAK ]
gap> Print("HK-rules:\n"); DisplayAsString( HKrules(dcrws4), alph4e, true );
HK-rules:
[ HbK, HAK ]
[ HbAbK, HAbAK ]
[ HbAbAbK, HAbAbAK ]
[ HbAbAbAbK, HAbAbAbAK ]
[ HbAbAbAbAbK, HAbAbAbAbAK ]
[ HbAbAbAbAbAbK, HAbAbAbAbAbAK ]
[ HbAbAbAbAbAbAbK, HAbAbAbAbAbAbAK ]

```

### 2.4.3 WordToString

- ▷ WordToString(*word*, *alph*) (operation)
- ▷ IdentityDoubleCoset(*dcrws*) (operation)

The NextWord operation (see 2.1.2) may be used to find normal forms of increasing length for double coset representatives. In the example below a limit of 50,000 (for the number of words tested) is specified since the 29 numbers of words tested can be shown to be:

```
[ 1, 1, 6, 9, 12, 4, 91, 12, 153, 12, 192, 52, 1435, 192, 12, 2457, 192,
12, 3072, 820, 22939, 3072, 19, 12, 39321, 3072, 192, 12, 49152 ]
```

Example

```

gap> idc := IdentityDoubleCoset( dcrws4 );
m1*m2
gap> ## List of the next 29 normal forms for double cosets:
gap> L4 := NextWords( dcrws4, idc, 29, 50000 );
gap> DisplayAsString( L4, alph4e, true );

```

```
[ HAK, HaK, HAbK, HbAK, HABAK, HAbAK, HABabK, HAbAbK, HbAbAK, HbaBAK, HABaBAK,\
  HAbAbAK, HABaBabK, HAbABabK, HAbAbAbK, HbAbAbAK, HbaBAbAK, HbaBaBAK, HABaBaBA\
  K, HAbAbAbAK, HABaBaBabK, HAbABaBabK, HAbAbABabK, HAbAbAbAbK, HbAbAbAbAK, HbaB\
  AbAbAK, HbaBaBAbAK, HbaBaBaBAK, HABaBaBaBAK ]
gap> w := NextWord( dcrws4, L4[29] );;
gap> Print( w, "\n" );
m1*(m3*m6)^4*m3*m2
gap> s := WordToString( w, alph4e );;
gap> Print( s, "\n" );
HAbAbAbAbAK
```



## Chapter 3

# Development History

### 3.1 Versions of the package

The first version of the package, written for GAP 3, formed part of Anne Heyworth's thesis [Hey99] in 1999, but was not made generally available.

Version 0.91 was prepared to run under GAP 4.4.6, in July 2005.

Version 0.94 differed in two significant ways.

- The manual was produced using the GAPDoc package.
- The test file `kan/tst/kan_manual.tst` set the `AssertionLevel` to 0 to avoid recursion in the Automata package.

Version 0.95, of 9th October 2007, just fixed file protections and added a CHANGES file.

Version 0.96 was required because the Kan website moved with the rest of the Mathematics website at Bangor.

Version 0.97, of November 18th 2008, deleted temporary fixes which were no longer needed once version 1.12 of Automata became available.

Version 1.01, of August 2011, included minor changes required for GAP 4.5. In particular, the test file `kan_manual.tst` was replaced by the pair `fulltest.tst` and `parttest.tst`.

Version 1.11, of December 2014 was required when the Kan website moved yet again. At the same time a bitbucket repository for the package was started.

Kan became an accepted GAP package in May 2015.

Version 1.28, of May 2017, saw a great many changes to the examples, with the various rewriting systems used to perform reduction of words to reduced form.

### 3.2 What needs doing next?

There are too many items to list here, but some of the most important are as follows.

- Implement iterators and enumerators for double cosets.
- At present the methods for `DoubleCosetsNC` and `RightCosetsNC` in this package return automata, rather than lists of cosets or coset enumerators. This needs to be fixed.
- Provide methods for operations such as `DoubleCosetRepsAndSizes`.
- Convert the rest of the original GAP 3 version of Kan to GAP 4.

### 3.2.1 DoubleCosetsAutomaton

- ▷ `DoubleCosetsAutomaton(G, U, V)` (operation)
- ▷ `RightCosetsAutomaton(G, V)` (operation)

Alternative methods for `DoubleCosetsNC(G,U,V)` and `RightCosetsNC(G,V)` *should be* provided in the cases where the group  $G$  has a rewriting system or is known to be infinite. At present the functions `RightCosetsAutomaton` and `DoubleCosetsAutomaton` return minimized automata, and Iterators for these are not yet available.

Example

```

gap> F := FreeGroup(2);
gap> rels := [ F.2^2, (F.1*F.2)^2 ];;
gap> G5 := F/rels;;
gap> genG5 := GeneratorsOfGroup( G5 );;
gap> a := genG5[1]; b := genG5[2];;
gap> U := Subgroup( G5, [a^2] );;
gap> V := Subgroup( G5, [b] );;
gap> L := [2,1,4,3];;
gap> rws5 := ReducedConfluentRewritingSystem( G5, L, "shortlex", 0, "aAbB" );;
gap> dc5 := DoubleCosetsAutomaton( G5, U, V );;
gap> Print( dc5 );
Automaton("det",5,"HKAaBb",[ [ 2, 2, 2, 5, 2 ], [ 2, 2, 1, 2, 1 ], [ 2, 2, 2, \
2, 3 ], [ 2, 2, 2, 2, 2 ], [ 2, 2, 2, 2, 2 ], [ 2, 2, 2, 2, 2 ] ],[ 4 ],[ 1 ])\
;;
gap> rc5 := RightCosetsAutomaton( G5, V );;
gap> Print( rc5 );
Automaton("det",6,"HKAaBb",[ [ 2, 2, 2, 6, 2, 2 ], [ 2, 2, 1, 2, 1, 1 ], [ 2, \
2, 3, 2, 2, 3 ], [ 2, 2, 2, 2, 5, 5 ], [ 2, 2, 2, 2, 2, 2 ], [ 2, 2, 2, 2, 2, \
2 ] ],[ 4 ],[ 1 ]);;

```

# References

- [BGHW06] R. Brown, N. Ghani, A. Heyworth, and C. D. Wensley. String rewriting systems for double coset systems. *J. Symbolic Comput.*, 41:573–590, 2006. 4, 6, 14
- [BH00] R. Brown and A. Heyworth. Using rewriting systems to compute left kan extensions and induced actions of categories. *J. Symbolic Comput.*, 29:5–31, 2000. 4
- [GH17] S. Gutsche and M. Horn. *AutoDoc - Generate documentation from GAP source code (Version 2017.09.15)*, 2017. GAP package, <https://github.com/gap-packages/AutoDoc>. 2
- [Hey99] A. Heyworth. *Applications of Rewriting Systems and Groebner Bases to Computing Kan Extensions and Identities Among Relations*. PhD thesis, University of Wales, Bangor, 1999. <http://www.maths.bangor.ac.uk/research/ftp/theses/heyworth.ps.gz>. 4, 17
- [Hor17] M. Horn. *GitHubPagesForGAP - Template for easily using GitHub Pages within GAP packages (Version 0.2)*, 2017. GAP package, <https://gap-system.github.io/GitHubPagesForGAP/>. 2
- [Lin91] S. Linton. Double coset enumeration. *J. Symbolic Comput.*, 12:415–426, 1991. 4
- [LN17] F. Lübeck and M. Neunhöffer. *GAPDoc (version 1.6)*. RWTH Aachen, 2017. GAP package, <http://www.math.rwth-aachen.de/~Frank.Luebeck/GAPDoc/index.html>. 2

# Index

DCrules, 14  
DisplayAsString, 8  
DisplayRwsRules, 6  
DoubleCosetRewritingSystem, 7  
DoubleCosetsAutomaton, 18  
  
example – free product, 7  
example – infinite rws, 13  
example – trefoil group, 10  
  
HKrules, 14  
Hrules, 14  
  
IdentityDoubleCoset, 15  
IsDoubleCosetRewritingSystem, 7  
IsWordAcceptorOfDoubleCosetRws, 9  
  
KBMagFSAtAutomataDFA, 13  
KBMagRewritingSystem, 13  
KBMagWordAcceptor, 13  
KnuthBendixRewritingSystem, 6  
Krules, 14  
  
NextWord, 7  
NextWords, 7  
  
PartialDoubleCosetRewritingSystem, 10  
  
ReducedConfluentRewritingSystem, 6  
ReducedForm, 6  
RightCosetsAutomaton, 18  
  
trefoil group, 10  
  
WordAcceptorByKBMag, 13  
WordAcceptorByKBMagOfDoubleCosetRws, 13  
WordAcceptorOfDoubleCosetRws, 9  
WordAcceptorOfPartialDoubleCosetRws, 10  
WordAcceptorOfReducedRws, 9  
WordToString, 15